空気抵抗のある運動

この章のテーマである「**微分方程式**」は、数学や物理だけの話ではなく、例えばウイルス感染の理論にも登場するなど、いまでは多くの学問分野で使われる基本的な「現象を方程式で表す方法」の一つとなっています。といっても、コンピュータを我々が個人でも持てるようになるまで「**微分方程式**」を解くというのは限られた人の仕事でした。コンピュータで数値的に解けるようになった今こそ、我々は、そのすごさを実感できるようになったといえるかもしれません。そういう意味でも今回の章は、興味のある研究テーマを数多く含んでいるといえるでしょう。

この講座の最初にやった**斜方投射の軌道**を覚えていますか。今思うと、「ちゃんとプログラムをしてやれば、あんなにフィルハンドルしてデータを角度毎に山のように出さなくてもよかったんじゃないの？」と言いたくなると思います。それは、君がコンピュータを使えるようになってきた証です。

ここでは、あの「**斜方投射**」から発展して、空気抵抗のある実際のボールの運動に近い状態を数値解として求めてみたいと思います。とはいっても、まず、空気抵抗のある自由落下から始めましょう。そして、今回は最終的な課題を先に与えておきたいと思います。それは、**花火**です。夜空に広がる花火の実際に近い状態をシミュレーションしてみることにします。この最終的な課題に向かって、途中までは引っ張って連れていきますが、最後は自分の力で山の頂上を目指してください。

それではほはじめましょう。



課題Ⅶ　花火シミュレーションの製作

大花火玉が鉛直上方に打ち上げられ最高点に達した瞬間、大花火玉は爆発し、大花火玉の中にあった 36 個の小花火玉がxy平面上に互いに 10° をなす角度で初速度 $v_0 = 50$〔m/s〕で 360°に広がった。広がる花火の時間毎の形状をアニメーションとしてシミュレーションせよ。なお、小花火玉には速さに比例した比例定数 k の空気抵抗が発生するとし、重力加速度をg=9.8 $\mathrm{m/s^2}$とする。kの値については、このテキストの最後のホームランボールの軌道を参考にせよ。

空気抵抗のある自由落下

通常の自由落下は、空気抵抗は無視できるとしているが、速さに比例する空気抵抗力kvがあったと仮定するとどうなるだろうか。

このときの運動方程式を書いてみると

$$ma = mg - kv$$

両辺をmで割って

$$a = g - \frac{k}{m}v$$

このとき　$\gamma = \frac{k}{m}$　とおいて自由落下の加速度は

$$a = g - \gamma v \quad \cdots ①$$

空気抵抗のある自由落下

空気抵抗力は、もちろん『力』です。速さに比例する力というのは、ピンとこない人もいるかもしれませんが、これは空気抵抗の理論から導かれた近似式です。

自転車を例にとればすぐわかるように、我々は空気の中で運動していますので必ず空気抵抗力を受けます。この抵抗力は、特に空気抵抗を受けるの断面積に大きな影響を受けるようです。

物体の形状と速さによっては、速さの二乗に比例するとした方がよく現実に合う場合もありますが、ここはより**線形**に近いものとしてとらえることにします。線形に近いというのは一次関数としてとらえるという意味です。

このとき空気抵抗力は kv〔**N**〕となりますので､比例定数k の単位は

k 〔**Ns /m**〕

ですが､〔**N**〕＝〔**kg・m /s²**〕より

k〔**kg /s**〕

となります。

すると、左の黒板に出てくる

$$\gamma = \frac{k}{m}$$

という量は〔**l /s**〕という単位になります。これを$\boldsymbol{\gamma}$（ガンマ）とおいておきます。

それでは、まず今回はエクセルのシートだけを使って、空気抵抗がある場合の、つまり現実に近い自由落下運動の $v-t$ グラフを描いてみることから始めたいと思います。

このテキストの初めにやったように、エクセルのシートそのものは、簡単にフィルハンドルを使って連続的に計算ができるように作られています。

そして次にプログラムコードを書いて同じことをしてみましょう。



図 5–5



図 5–6

SHEET 1 に計算

まず「Sheet1 にちょっと計算してみる」というスタイルでやってみます。これも大事なスキルでしたね。プログラムを組まなくても、エクセルの表計算は、計算やグラフを描くのに大変便利です。

まずエクセルを立ち上げ、名前を「**自由落下 Sheet1**」としておきます。

図 5–5 のようにまず定数と数値、そして単位をしっかり書きます。これはどんな時も忘れないように！タイトルは自分で工夫して書いてください。図 5–5 は数式をワードに書いてコピペしてちょっと広げてます。**γ＝k／m** はとりあえず 1 にしておいて、グラフができてから調整することにします。

時間 **t** はセル C12 で

=C11+D8

として D8 はキー F4 を押して絶対参照にしておきます。

=C11+D8

そしてセル C12 を**フィルハンドル**してとりあえず 7s 程度まで引っ張ります。結構ありますよね。

図 5–6 がキモです。左図の 0.01s 後の速さが計算させる**セル D1** は =D11＋E11＊D8

ここでキー F4 を押して D8 を絶対参照にして

=D11+E11*D8

とします。

ここで、もうわかっているとは思いますが、例えば D11 と書きたいときにはセルの D11 をマウスでクリックすれば D11 となるのでしたね。ここでまだ D11 といちいち書いている人は・・・古代人です。

この計算式は、等加速度運動の $v = v_0 + at$ が 0.1s 間だけ成立していると仮定していることになります。



次にセル E12 は　$a = g - \gamma v$ の式でこの 0.1s 間の加速度を計算します。そのときの v には左隣のセルの速さを使っています。ここが最も面白いところですね。この v 使っていいのお？という感じですがこれは、図 5-7 で示したように、加速度の縦のセルは、セルの二分の一だけ下がったところに実際のセルがあると考えると納得してもらえると思います。つまり 0〜0.01s の間は加速度 9.8m/s^2 ということです。あとはこのセル 12D と 12E を一緒に下にフィルハンドルします。

最後、図 5-8 のように$v-t$グラフを描いてみましょう。これは、説明入りませんね。うまくグラフが描けない人は仲間に教えてもらいましょう。どんなグラフになるのでしょうか。空気抵抗がないときは$\gamma=0$ですよね。いろいろ実験してみてください。

実際は加速度 a のセルは、半分だけ下がっていて、例えば 0.01～0.02s の間の加速度は9.702m/s^2と考えると計算式を入れやすくなる。



図 5-7





図 5-8



図 5-9

```
(General)                自由落下

    Dim dt, dv, dy, t, v, a, y As Double
    Dim n As Integer

Sub 自由落下()
'初期値の読み込み

    dt = Cells(3, 6)
    g = Cells(4, 6)
    γ = Cells(5, 6)
    v = Cells(8, 5).Value
    t = Cells(8, 4).Value
    y = Cells(8, 3).Value

'計算ループ

  For n = 1 To 1000

   dv = (g - γ * v) * dt
   dy = v * dt + 0.5 * dv * dt

    t = t + dt
    v = v + dv
    a = g - γ * v
    y = y + dy

  'セルに順番に書いていく

    Cells(8 + n, 4) = t
    Cells(8 + n, 5) = v
    Cells(8 + n, 6) = a

    DoEvents

  'y をマイナスにして5の倍数を拾う

      If n Mod 5 = 0 Then
         Cells(8 + n, 3) = -y
      End If
```

図 5-10

プログラムを組む

さて次にプログラムを組んで同じことをやってみましょう。新しいエクセルを立ち上げます。ファイル名は「**自由落下プログラム**」としておきましょう。

図 5-9 が**実験ラボ**です。**図 5-10** が**プログラムコード**になります。

自分でコードを書いてみようと思う人は、はじめはなかなか動かなくていやになりますが、走り出したときにはちょっとした仲間ができた喜びがあります。そして、それ以上に新しい言語、コンピュータと話す言語を手に入れた不思議な感覚があります。この BASIC 以外にも C++（シープラ）や Python（パイソン）などコンピュータと話す言語はありますが、便利なことに、一つだけ覚えれば、あとは似たようなものです。プログラミングは制御文（もし○○なら）がすべてといってもいいのです。その形式はどれも似たり寄ったりです。

今回のプログラムは制御文があまり登場してきませんね。ちょっとコンピュータが「考えてる」感じが薄いかもしれません。計算をやらされまくってる感じです。



図 5-11



図 5-12



図 5-13

実験オイラー法の精度

　オイラー法による数値計算で曲線を求めると、知ったかぶりの奴が必ず「オイラー法？精度悪いでしょう。なんで、ルンゲクッタ法でしないの？」とか言ってくるという話はすでにしました。本当に精度がいまいちなのか、この実験ラボで実験してみることにします。

　微分方程式は「**傾き**」つまり $\frac{dx}{dt}$ や $\frac{dv}{dt}$ しか曲線の情報を与えませんから**図 5-11** のように、求める曲線は実は折れ線になります。この実験をやってみましょう。

準　備 1

　まずグラフの座標軸が固定されているか確認しましょう。**図 5-12** のようにグラフの時間軸の目盛りをクリックしてメニューを出し「**軸の書式設定**」を選びます。**図 5-13** のように最小値と最大値を手動で書いて リセット になるようにしますが 0.0 はなかなかいうことを聞いてくれませんね。僕は 0.5 とかわざと入れたあと 0.0 にします。するとリセットになります。

　縦軸も**図 5-13** のようにします。ここでは速さの最大値を 12m/s にしています。実は、雨の水滴のスピードは測定されていておよそ 10m/s 程度の等速です。



図 5-14



実験

γ=1.2 と少し抵抗を大きくしてグラフを下に下げ、dt を 1.2s から 0.1ｓずつ減らしながら、それぞれの場合どのような曲線になるのかを観測せよ。

　さらにｄｔが 0.1s から 0.05s,0.01s,0.005s,0.001s と小さくしていくと、曲線はどのようになっていくか観測せよ。

図 5-15

準備２

　グラフを折れ線にします。図 5-14 上のようにグラフの曲線を右クリックしてメニューを出し「**系列グラフの種類の変更**」を選びます。

そこで図 5-14 下「**散布図**」の**折れ線の図**を選びます。

実 験

① **γ=1.2 にしてグラフを下にちょっと下げます。**

② 図 5-15 のように dt の値を 1.2 s というとんでもなく大きく荒い値を入れてみましょう。オイラー法の本質が分かります。1.3 以上入れるとグラフからはみ出しますのでやめたほうがいいです。

③ dt=1.2s から値を 0.1s ずつ小さくしていって、折れ線がどのようになっていくか観測してください。

③ 次にベストの dt の値を探します。ベストというのは、これ以上小さくしても、ほとんど曲線の形状は変わらないという dt の大きさのことです。むやみやたらに小さくすればいいわけでもないことが分かります。例えば 0.001s というような現実あり得ない時間間隔は、計算にやたら時間がかかりますが、曲線の形状はどうでしょうか。ここにもオイラー法の本質が見えてくるはずです。

参考

　オイラー法が苦手な微分方程式は、sin やその兄弟の cos が出てくるときです。突然増加が減少に転換しますので、オイラー法ではついていけずそのピークが高くなってしまいます。でも、手軽で単純な理論で扱いやすい、かわいいオイラー法をこれからかわいがってください。

「ルンゲクッタのほうが精度がいいんじゃないの？」

「オイラーのすごさも、おいら知ってるんで・・・」



図 5-16

実験２　落下の観測

　図 5-16 を見てください。自由落下の実験ラボを改良しました。プログラムを組むと、このように少しずつ改良できるところが魅力です。

　さて、みなさんどんなところが変わっているかわかりますか。

　図 5-16 では、γ=0 になっていて自由落下になっています。そして、その自由落下の実際の様子が、右にグラフとして描かれていますね。

　どうやって書いているのでしょう。左側を見ると x と y の欄が増えていて、y の欄には 0.5s おきにデータが書かれてあります。どうもこれがグラフに反映されているようです。では、この 0.5s おきにデータを書かせるってどうすればいいのでしょう。

　左のコードが、図 5-16 の計算をするために改良したプログラムです。次のページにまたがっていますので注意してください。

　しっかり読んでいくと、少しずつ変わっていますね。大きく変わったところを２か所だけ説明したいと思います。

```
Dim dt, dv, dy, t, v, a, y As Double
Dim n As Integer

Sub 自由落下()

  '初期値の読み込み
      dt = Cells(3, 6)
      g = Cells(4, 6)
      γ = Cells(5, 6)
      v = Cells(8, 5).Value
      t = Cells(8, 4).Value
      y = Cells(8, 3).Value
  '計算ループ
     For n = 1 To 1000
          dv = (g - γ * v) * dt
          dy = v * dt + 0.5 * dv * dt
            t = t + dt
            v = v + dv
            a = g - γ * v
            y = y + dy
         'セルに順番に書いていく
          Cells(8 + n, 4) = t
          Cells(8 + n, 5) = v
          Cells(8 + n, 6) = a
          DoEvents
        'n が 5 の倍数のときだけセルにデータを書く
         If n Mod 5 = 0 Then
              Cells(8 + n, 3) = -y
         End If
     Next n
End Sub
```

```
Private Sub CommandButton1_Click()
  自由落下
End Sub
```

```
Private Sub CommandButton2_Click()
 Range(Cells(9, 3), Cells(10000, 6)).Clear
End Sub
```

dy はどうやって計算するのか？

dv = (g - γ * v) * dt
dy = v * dt + 0.5 * dv * dt
中略
y = y + dy



図 5-17

コードの説明

プログラムには dy を求める式が追加されました。これはボールが座標 y まで進んだ後、時間が dt 経った時、進んだ距離を dy とすると、この dy はどうやって計算すればいいかという式です。

説明のために**図 5-17** を描きました。v-t グラフの面積は距離になることは知っていますか。この図はそのことが基本になっています。

「面積 y」まではボールが y〔m〕進んできたことを示しています。その瞬間速さが v だったとします。次に dt だけ時間がたつと

面積 1 と面積 2 を足した分

だけ進んでいることになりますね。この面積 1 は

$$\frac{1}{2}dt * dv$$

となりますし、面積 2 は

$$v * dt$$

となりますね。これを足したものが dt の時間で進んだ距離になりますので

$$dy = v * dt + \frac{1}{2}dt * dv$$

ということになります。

すると、t+dt 時間がたつと進んだ距離は y+dy となりますので、プログラムとしては、これを新しい y とするという意味で

y ⇐ y + dy

この ⇐ はキーボードにはありませんので＝を使って

y = y + dy

と書くわけです。

```
Mod ってなに？
    'n が 5 の倍数のときだけセルにデータを書く
        If n Mod 5 = 0 Then
            Cells(8 + n, 3) = -y
        End If
```

n	x	
0	1	
1	1	
2	1	
3	1	
4	1	
5	1	-
6	1	
7	1	
8	1	
9	1	
10	1	
11	1	
12	1	
13	1	
14	1	
15	1	-1
16	1	



図 5-18

MOD って何？

プログラムコードを読んでいてもう一つ気づくのは、左の黒板に書いたコードです。

IF 文は、これまで何回も出てきましたのでまあいいとして、

n Mod 5 = 0

というのはどういう意味でしょうか。これはかけたり割ったりする四則演算の中のルールに登場する

n を 5 で割った余りが 0

という意味です。言い換えると

n が 5 の倍数の時

ということです。

つまりこのコードは

もし n 回目の n が 5 の倍数だったらセル(8+n,3) のとこに -y の値を入れといてくれない！？

そうじゃなかったら、なにもしなくてもいいんだけどさあ、

という意味です。なぜこんなことをしたのかというと、図 5-18 の右にあるようなボールの落下の軌跡を一定の時間間隔で撮影したものを図にすると、ボールの速さがビジュアル的にみてわかるようになるからです。

図は「自由落下」のときです。0.5s 間隔でフラッシュ撮影したようにボールの軌跡が出てきます。時間がたつほどにボールが速くなっている様子が分かりますね。

それでは空気抵抗があるときはどのようになるか、自分で実験してみましょう。

Mod … 花火のシミュレーションにも使えそうですね。

空気抵抗のある斜方投射

運動方程式は

x軸方向　$ma = -kv\cos\theta$

y軸方向　$mb = -mg - kv\sin\theta$

両辺をmで割って

x軸方向　$a = -\gamma v\cos\theta$

y軸方向　$b = -g - \gamma v\sin\theta$

$\gamma = \frac{k}{m}$ ， $v\cos\theta = v_x$ ， $v\sin\theta = v_y$ **とおく**

$$\frac{dv_x}{dt} = -\gamma v_x$$

$$\frac{dv_y}{dt} = -g - \gamma v_y$$

この２つの微分方程式を使ってオイラー法でプログラムを組んでいく。



図 5-19

空気抵抗のある**斜方投射**

「**花火課題**」解決に向けての最終タスクになりました。「**空気抵抗のある斜方投射**」をオイラー法を使って解くことにしましょう。

この講座の最初に砲丸投げの最適角度の計算をしました。砲丸のように質量 m が 4〜7kg と大きいものは

$$\gamma = \frac{k}{m}$$

においてγ→小 となって空気抵抗を無視することができます。だからあの砲丸投げの最適角は有効な理論なのです。

それでは、野球などのボールはどうなのでしょう。質量は 150g を切ります。このときの空気抵抗はボールの運動に影響してきそうです。実際のホームランボールは、最近は測定器が発達して、ボールの回転数まで測定できるようになりました。ハイスピードカメラとコンピューターを使って分析できるようになったのです。いち早く MLB（アメリカ大リーグ）がこの分析とデータを使った野球をするようになっています。エンジェルスの大谷翔平がホームランを打った時に、初速度 50m/s 程度で135m飛んだというニュースを耳にしました。これはγを決定するうえで重量なデータです。

さて、ボールに飛行中に加わる力は空気の抵抗力 kv と重力 mg です。これを水平 x 軸と鉛直 y 軸に分解すると、**図 5-19** のようになります。これを使って運動方程式を書くと、上の黒板のようになります。

結局、この鉛直方向と水平方向の微分方程式を組み合わせて解いていけばよいということになります。お互いの式は独立しているので、これまでやってきたオイラー法で解くことができます。



図 5-20

```
Dim t, dt, v0, vx, vy, dvx, dvy, dx, dy, x, y, γ, θ As Double
Dim n As Integer

Sub 空気抵抗のある斜方投射()
   '初期設定
        dt = Cells(2, 8)
         g = Cells(4, 8)
         γ  = Cells(3, 8)
         θ = Cells(5, 10)
        v0 = Cells(6, 8)
          t = 0
          x = 0
          y = 0
        vx = v0 * Cos(θ)
        vy = v0 * Sin(θ)

   'オイラー法による計算
      For n = 1 To 100
          dvx = -γ  * vx * dt
          dvy = -(g +  γ  * vy) * dt
          vx = vx + dvx
          vy = vy + dvy
          dx = vx * dt + 0.5 * dt * dvx
          dy = vy * dt + 0.5 * dt * dvy
           t = t + dt
           x = x + dx
           y = y + dy

         '計算結果をシートに書く
          Cells(10 + n, 3) = t
          Cells(10 + n, 4) = x
          Cells(10 + n, 5) = y
      Next n
End Sub
```

```
Private Sub CommandButton1_Click()
   空気抵抗のある斜方投射
End Sub
```

実験ラボをつくる

図 5-20 を見てください。これが空気抵抗のある斜方投射用の実験ラボです。

初速度 50m/s (180km/h)で大谷翔平レベルで打ち出し 110m ほど飛んでいます。この時仰角 50°で抵抗係数は **γ**=0.2 としました。仰角はもっと小さく 30° 台でしょう。

実験ラボが完成したら、これらを調整して 135m ぐらい飛行する **γ** の値を決めましょう。これで現実のメジャーリーグでの軌道シミュレーションが可能になります。

さらに実際の花火プログラムでもその抵抗係数 **γ** が使えそうですね。スタートで花火が上がって行き上空で花火が開く、そして消えてなくなるというようなプログラムができたら最高ですね。そんなレベルがつくれる人はトップレベルですから、ぜひ「きみろん」の研究テーマとして発表してください。ポスターセッションで、ポスターの前にパソコンを置いて発表するのです。みんなの注目を浴びることは間違いないでしょう。